

## COMPUCOLOR CORPORATION ANNOUNCES MAJOR PRICE CUTS

Readers --

Our big news this month is that prices have been reduced on a number of options and accessories. We are able to make these price cuts because of reduced memory costs, lower keyboard costs, and high sales volume. These price changes will make some of the most popular accessories more easily affordable. Those of you with Models 3 or 4 will note on the enclosed order blank that the price of the additional 16K of RAM has been cut in half. You can expand your machine's capabilities at a new low price. This is especially helpful for those of you with the Model 3 COMPUCOLOR II, since you can now update your machine just in time to figure your taxes with the new 16K TAX '78 Sof-Disk Album. The price cut that will undoubtedly be the most well-received is the one involving our formatted blank Sof-Disks. The price has been reduced from 2 for \$20.00 to 2 for \$9.95 -- a reduction of over 50%! We hope this will encourage you to come up with some clever programs.

\*\*\*\*\*

## This Issue's MENU

Editor's Letter .....	1
KEEPING IT SIMPLE	
Dotted Lines.....	2
SOFTWARE EXPLAINED	
Hangman Modifications .....	3
DISK DUP .....	4
ADVANCED APPLICATION	
The MENU Program.....	6
NEW PRODUCTS .....	7
COMING UP .....	7

\*\*\*\*\*

We know that some of you are about ready to make a contribution to the Users Group Software Files. We've had several promises from readers who will be sending in their best efforts. It looks like the Users Group is going to have a good selection of software for trade in just a few months, so think about sending in your programs for review. The Software Files will be a great way to find out how others are making use of their COMPUCOLOR II's. From our market surveys, it seems that most of you are interested in games of skill and financial applications. (What a sophisticated group!) So keep that in mind when you decide where to concentrate your efforts. Some of our Californian friends are asking about real time Space War games. We hope someone out there will come up with a spectacular one. Remember that an exceptional program can earn you up to \$1000. Due to our mail volume, we will probably be starting a 'Readers Talk Back' sort of column next month. Some of you are prolific correspondents, but others of you are silent! All letters are welcome, so take a few minutes and let us know what you're thinking.


## KEEPING IT SIMPLE

### Dotted Lines

This month's topic was suggested by Barbara Freistuhler, of Los Angeles, California, who asked for a way to draw a dotted line between two points instead of the solid line drawn by the vector plot mode. Because the COMPUCOLOR II does not have infinite resolution, sloping lines can sometimes be irregular in appearance. Using a dotted line can create a neater, more concise display.

The program below allows you to specify two endpoints and an increment factor. The following explanation will illustrate the development of a simple but useful algorithm and the resulting program.

The idea behind the program is shown below. DX is the distance of the base line and DY the length of the side line formed by drawing vectors from the endpoints of a sloping line. Points are selected on the longest base line, and the slope determines where to plot the corresponding point on the line to be drawn.



This idea translates into the usable program below:

```

5 PLOT 12
10 INPUT "X0,Y0,X1,Y1,INC";X0,Y0,X1,Y1,INC
20 GOSUB 1000
30 INPUT "AGAIN?";A$
40 GOTO 5

1000 DX = X1 - X0: DY = Y1-Y0: PLOT 2
1010 IF DX=0 AND DY=0 THEN 1200
1020 IF ABS (DY) > ABS (DX) THEN 1100
1030 SL = DY/DX
1040 FOR X=X0 TO X1 STEP INC * SGN (DX)
1050 PLOT X, Y0+(X-X0) * SL+.5
1060 NEXT X
1070 GOTO 1200
1100 SL =DX/DY
1110 FOR Y=Y0 TO Y1 STEP INC * SGN(DY)
1120 PLOT X0+(Y-Y0) * SL+.5, Y
1130 NEXT Y
1200 PLOT X1,Y1,255:RETURN

```

Line 5 erases the screen. Line 10 uses the INPUT statement to let the user define the line's endpoints. Line 20 sends the program to the subroutine that calculates and plots the dots. Line 1000 calculates the length of the two lines that form the base and sides of the triangle pictured above. PLOT 2 enters plot mode. Line 1010 tests for the two endpoints being identical. If this is the case, control passes to line



1200, and only one point is plotted. Line 1020 selects the triangle base line from which to project the dots. If the triangle's side is longer than its base (i.e.,  $|\text{slope}| > 1$ ), calculations are made at line 1100. Line 1030 calculates the slope by the 2-point method. (You recall this formula from Algebra I!) Line 1040 begins the FOR loop that actually draws the points, and line 1060 closes this loop. Line 1100 begins the calculation and plotting sequence for lines with the absolute value of the slope being  $>1$ . The program always plots both endpoints of the line. When the plot mode is exited and control is passed back to the main body of the subroutine, line 30 asks if another run is desired. Notice that this inquiry serves no function other than to delay erasing the screen to give you time to examine the dotted line. The program could easily be changed to terminate if the response to the question were 'NO'.

This formula for dotted lines can be incorporated into a program with the proper endpoints and increment value to achieve any desired display.

## SOFTWARE EXPLAINED

### Hangman Modifications

The market surveys tell us that many of you are involved with education in some way -- getting it or giving it -- at primary, secondary, or University levels. Some teachers have mentioned that being able to modify the Hangman program to insert a special word list would be useful. You can use words of your choice for vocabulary and spelling drills. Others can modify word lists to add a few favorite puzzlers. This month we'll show you how to select any words you choose and enter them into Hangman's lexicon.

The program below allows you to create a new LIB (library) file for HANGMAN. Enter the program as it appears below, and save it on disk for convenient access. When you RUN the program, it will guide you through the required word entry.

```

90 REM *****
91 REM
92 REM PROGRAM FOR CREATING USER DESIGNED FILE(S) FOR HANGMAN
93 REM
94 REM *****
95 REM
96 REM          THIS PROGRAM IS SET UP FOR WORDS NO LONGER
97 REM          THAN 16 CHARACTERS IN LENGTH
98 REM
99 REM *****
100 REM -- TO CREATE OR UPDATE A FILE
110 INPUT "UPDATE OLD LIBRARY (0) OR CREATE NEW LIBRARY (1) ?";A
120 IF A<>0 AND A<>1 THEN GOTO 28,11:GOTO 110
130 ON A+1 GOTO 210,140
140 INPUT "NEW LIBRARY NAME?";L$
150 INPUT "WORDS IN FILE?";W
160 FILE "N",L$+ ".LIB",W,16,8
170 FILE "R",1,L$+ ".LIB",1
180 PRINT "FILE ";L$;".LIB CREATED FOR ";W;"WORDS."
190 PUT 1,1;1,W

```

```

200 GOTO 250
210 INPUT "OLD LIBRARY NAME? ";L$
220 FILE "R",1,L$+".LIB",1
230 GET 1,1;N,W
240 PRINT "FILE ";L$;".LIB CONTAINS";N-1;"OUT OF";W;" WORDS."
250 INPUT "ENTER NEW WORD OR '0' ? ";W$
260 IF W$="0" THEN 300
270 N=N+1
280 PUT 1,N;W$[16]
290 GOTO 250
300 INPUT "END RUN (0), OR LIST WORDS (1) ? ";A
310 IF A<> 1 AND A<>0 THEN PLOT 28,11:GOTO 300
320 ON A+1 GOTO 390, 330
330 IF N=1 THEN PRINT "NO WORDS ON FILE.": GOTO 300
340 FOR I=2 TO N
350 GET 1,I;W$[16]
360 PRINT W$
370 NEXT I
380 GOTO 300
390 PUT 1,1;N,W
400 FILE "C",1
410 END

```

After you use this program, you will notice that a new file appears on the directory; for example, WORD.LIB. In order for the Hangman program to access this new file, a few lines in the Hangman program must be changed, as described below:

```

667 GOSUB 4000
668 (delete)
add:
4000 REM
4010 IF N8$="E" THEN FILE "R",1,"EASY.LIB",1
4020 IF N8$="H" THEN FILE "R",1,"HARD.LIB",1
4999 RETURN

```

For every new .LIB file you create, you must add a line to the subroutine that allows for access. For example:

```

4030 IF N8$="W" THEN FILE "R",1,"WORD.LIB",1

```

will allow the program to access a new .LIB file -- WORD.

## DISK DUP

On some of the MATHTUTOR Sof-Disks there is a program called DUP. The MENU program does not access it, but many of you have found it on the directory and made use of it. It allows you to duplicate a Sof-Disk without having a second disk drive. This is very handy, as it's always a good policy to have a back-up copy of anything you're working on. For those of you who do not have a copy of DUP, a source listing is printed below. Just enter it into the COMPUCOLOR II and save it on a disk. While it runs, it gives you instructions on how to use it.



```

0 REM
1 REM ***** DUP.BAS;01 *****
2 REM
5 CLEAR 100
10 T = PEEK(32940) + 256 * PEEK(32941)
30 P = PEEK(32984) + 256 * PEEK(32985)
50 A = INT ((P+200)/256): B = (P+200) - A * 256
70 C = INT (T/256): D = T - 256 * C
100 A = A+1: B = 0
102 C = C-2: D = 255
105 P = A*256+23: REM P NOW POINTS TO FIRST ATTRIBUTE BYTE
110 X = A: GOSUB 900: A$ = H$
120 X = B: GOSUB 900: A$ = A$+H$
125 A$ = A$+"-"
130 X = C: GOSUB 900: A$ = A$+H$
140 X = D: GOSUB 900: A$ = A$+H$
160 I = 0
165 PLOT 6,2: INPUT "SOURCE DISK";Z: PLOT 27,4
166 X = I: GOSUB 900
170 PRINT "READ ";H$;" ";A$
172 PLOT 27,27: IF I=0 THEN GOSUB 800
173 PLOT 6,6: INPUT "DESTINATION DISK";Z: PLOT 27,4
174 PRINT "WRITE ";H$;" ";A$
190 I = I+INT ((T-P)/128)
194 IF I=>Q THEN 1000
195 IF I+INT ((T-P)/128) > 400 THEN I = 400-INT ((T-P)/128)
197 PLOT 27,27
200 GOTO 165
500 GOTO 1000

800 REM
810 INPUT "ARE THERE ANY ITEMS TO BE COPIED NOT SHOWN IN
      THE DIRECTORY? ";Q$
820 IF LEFT$(Q$,1) = "Y" THEN Q=400: RETURN
830 Q = PEEK(P+11) +256 * PEEK(P+12)
850 IF PEEK(P) = 1 THEN P = A * 256: RETURN
860 P = P+21
870 IF INT (P/128) = P/128 THEN P=P+2
880 GOTO 830
890 RETURN

900 REM X=VALUE TO BE HEXED
905 H$=""
910 X1 = INT(X/16): X2 = X-16 * X1
920 IF X2<10 THEN X2=X2+48: GOTO 940
930 X2 = X2+55
940 IF X1<10 THEN X1=X1+48: GOTO 960
950 X1 = X1+55
960 H$ = H$ + CHR$(X1) + CHR$(X2)
965 IF X1<71 THEN RETURN
970 X1 = X1-71
980 H$ = "1"
990 GOTO 940

1000 REM
1010 END

```

## ADVANCED APPLICATION

### The MENU Program

You may have noticed that the SAMPLER Sof-Disk Album has two ways of using the programs on that Album. You may access them individually, or you may type in 'ALL' and have the program cycle through a demonstration of each program. When 'ALL' is entered to the SAMPLER MENU, the program enters a demonstration mode where each program on the disk is run consecutively. When the last program is run, the program goes back to the first program and repeats the sequence. Since each program is loaded separately, one after the other, it may seem odd that the program knows to stay in the demonstration mode and not ask for user input. You may also wonder how the program is able to load and cycle through the programs in numerical order.

Actually, this is accomplished through a fairly simple procedure. It is done by using one byte of RAM which is outside the BASIC user memory. In this instance, location 32936 is used. If you look at the listing of MENU, you will see that when it RUNs, line 4 POKES a 0 (zero) into location 32936. Line 4000 POKES a 1 (one) into this address if 'ALL' is entered. If any other character(s) is entered, a zero is POKEd into this location.

When each program runs, it PEEKs at the value in 32935 to determine if it should run as a demonstration program (PEEK (32936) <> 0). (See BANDIT lines 126, 326, 465, 965 and 1040 as an example.) When the program returns to MENU, the statement:

```
LOAD"MENU":RUN 5
```

is used. This statement causes MENU to be loaded with execution starting at line 5. If this were not done, then line 4 would also be executed and the value in 32926 set back to zero. MENU checks the value in 32926. If the value is zero, the MENU program is executed in the normal way. If not zero, then the program is in the demonstration mode, in which case the value is incremented by one and the next program loaded.

## NEW PRODUCTS

### Keyboard and Pin Cushion

A horizontal pin cushion kit is available by writing to us. It is possible for you to install it yourself, but we do recommend that your dealer perform the installation. This kit will largely eliminate the side curvature of the screen display and make the display appear more rectangular.



A new keyboard case is available that replaces the old square-cornered model. The cost is \$50.00, and it includes the up arrow cursor control key and switch, and the new Thin-Line case (available in tan only). For installation, the board edge connector must be shortened by .25" to fit the new case. This is best done with a fine file or fine grinder wheel.

#### COMING UP

Call for Articles  
Reader Exchange  
Personal Finance  
Bio-Fix

Several readers have requested COMPUCOLOR II literature to give to friends or co-workers. We have included two of our new brochures with this issue.

When December ColorCue went to print, the last line of the first page was inadvertently cut off. It should have read " and is indispensable for using color and graphics." Our apologies.

The following is a letter that was sent in by one of our users to Personal Computing Magazine. We are pleased at such a positive response. It is our policy to handle your problems as effectively as possible. Let us know if we can help you out!

Dear Editor,

As beginners in computers, we enjoy your magazine immensely. To help other beginners who often are disillusioned by lack of info, we want to pass on our enthusiastic endorsement of the COMPUCOLOR II to people. They have an excellent product at a good price and their customer service has been unbelievable! It's been many years since we've encountered people at a company who are so helpful and understanding. They really listen and understand problems and suggestions. We received our computer BEFORE the promised date! They have a Hot Line and we've been helped time and again by both hardware and software people who understand questions the first time and give answers promptly.

W.A. Shanks  
Miller Place, NY

# COMPUCOLOR CORPORATION ORDER FORM

PART #	DESCRIPTION	QUANTITY	PRICE	SUB-TOTAL
A8CC23	Compucolor II Model 3, 8K	_____	\$ 1495.00	\$ _____
A8CC24	Compucolor II Model 4, 16K	_____	\$ 1695.00	_____
A8CC25	Compucolor II Model 5, 32K	_____	\$ 1995.00	_____

101001 Standard Keyboard, 72 Key )  
 990002 Sof-Disk - Sampler ) INCLUDED WITH ALL UNITS  
 999207 Instruction Manual )

## OPTIONAL KEYBOARDS (To be sent in place of 101001 Standard Keyboard)

101002	CCN Keyboard, 101 Keys	_____	\$ 135.00	_____
101000	CCNF Keyboard, 117 Keys	_____	\$ 200.00	_____

## ADDITIONAL OPTIONS

A8CD11	Add on Single Disk 51.2K Bytes	_____	\$ 400.00	_____
010044	RS232 Adapter CA Assy.	_____	\$ 50.00	_____
100986	16K RAM Module (Models 3 & 4 only)	_____	\$ 375.00	_____
999207	Instruction Manual	_____	\$ 3.00	_____
999208	Maintenance Manual	_____	\$ 50.00	_____
999209	Programming and Reference Manual	_____	\$ 25.00	_____
990001	Sof-Disk Formatted (Twin Pack)	_____	\$ 9.95	_____
990002	Sof-Disk Sampler	_____	\$ 19.95	_____
990003	Sof-Disk Hangman	_____	\$ 19.95	_____
990004	Sof-Disk Othello	_____	\$ 19.95	_____
990005	Sof-Disk Math Tutor	_____	\$ 19.95	_____
990006	Sof-Disk Chess (16K)	_____	\$ 19.95	_____
990007	Sof-Disk StarTrek	_____	\$ 19.95	_____
990008	Sof-Disk BlackJack	_____	\$ 19.95	_____
990009	Sof-Disk Cubic Tic-Tac-Toe (16K)	_____	\$ 19.95	_____
990010	Sof-Disk Personal Finance Vol. 1	_____	\$ 19.95	_____
990011	Sof-Disk Personal Finance Vol. 2	_____	\$ 19.95	_____
990012	Sof-Disk Bonds (16K)	_____	\$ 19.95	_____
990013	Sof-Disk Equity (16K)	_____	\$ 19.95	_____
990014	Sof-Disk Assembler(with Manual-16K)	_____	\$ 24.95	_____
990015	Sof-Disk Text Editor(w/Manual-16K)	_____	\$ 24.95	_____
990016	Sof-Disk Personal Data Base	_____	\$ 29.95	_____
990017	Sof-Disk Income Tax '78 (16K)	_____	\$ 29.95	_____
010047	Spare Parts Kit - Analog	_____	\$ 60.00	_____
010048	Spare Parts Kit - Digital	_____	\$ 100.00	_____
999001	Freight	_____		_____

(\$45 per computer, plus 5% on all ADDITIONAL OPTIONS)

(unless otherwise specified, shipment is by Air Freight)

California, Georgia, and New York Residents must add sales tax \_\_\_\_\_

TOTAL \$ \_\_\_\_\_

Special Shipping Instructions, if any \_\_\_\_\_

NAME \_\_\_\_\_ ADDRESS \_\_\_\_\_

TELEPHONE # (HOME) \_\_\_\_\_

(BUSINESS) \_\_\_\_\_ ZIP \_\_\_\_\_

TERMS--Cash with order.

WARRANTY--90 Days Parts and labor at factory. All warranty repairs are to be shipped to the factory COLLECT. All returns to the factory must have a Return Authorization Number, which is obtained by calling (404) 449-5961. All Computer Programs sold on an "AS IS" basis without warranty.

ORDER MUST BE SIGNED FOR ACCEPTANCE

CC217-179

SIGNATURE \_\_\_\_\_